

Funkamateure haben den Maidenhead- bzw. QTH-Locator entwickelt, um die geografische Position auf der Erde (Längen- und Breitengrad), das QTH, auf eine Buchstaben- und Ziffernfolge zu verdichten.

Beispiel Reichstagsgebäude Berlin, Sitz des Deutschen Bundestages

52,518591° N (nördliche Breite, Latitude), 13,376109° E (östliche Länge, Longitude)

verdichtet auf JO62qm, um 2 Stellen genauer: JO62qm54.

Link: <https://www.fun-funk.net/aprs4web/locator/?qth=JO62qm54>

Während die geografische Breite seit Jahrhunderten auf den Äquator bezogen war, 0...+90° nördliche Breite und 0...-90° südliche Breite, gab es national unterschiedliche Festlegungen des Nullmeridians (Längengrad 0). Die Festlegung des Nullmeridians erfolgte 1884 auf der internationalen Meridian-Konferenz in Washington. Man folgte dort der damaligen Seemacht Großbritannien mit der Festlegung der Länge 0° auf den Ort der Sternwarte Greenwich.


Die geografische Länge wird vom Nullmeridian positiv nach Osten (0...+180°, 0...180° E (East) und negativ nach Westen (0...-180°, 0...180° W (West) gezählt.

Der Längengrad +180° = -180° verläuft etwa durch die Wrangelinsel im Ostsibirischen Meer.

Das derzeit gültige Referenzsystem ist das World Geodetic System 1984 (**WGS84**). Dessen Internationaler Referenzmeridian (IRM) liegt etwa 100m östlich der alten Bodenmarkierung in Greenwich.

Damit wird auch die Uhrzeit auf den Nullmeridian bezogen, seinerzeit GMT (Greenwich Mean Time), heute UTC (Coordinated Universal Time), beide nur annähernd identisch.

Wenn das Ellipsoid der Erdoberfläche in eine Ebene aufgerollt wird, ergeben die Längen- und Breitengrade nach WGS84 ein rechteckiges Raster. Zur Berechnung des Locators werden unterschiedlich große Rechtecke vorgegeben, innerhalb derer mit jeweils verfeinerter Schrittweite die Geo-Position (Länge/Breite) lokalisiert wird.



Eingefügte Erdkarte: <https://de.wikipedia.org/wiki/QTH-Locator>

## 1. Größtfeld (engl. Field):

Geografische **Länge** (-180...+180°) in Abschnitten von **20°**,

ergibt 18 Abschnitte, gekennzeichnet mit A...R (Buchstaben 1 bis 18).

Geografische **Breite** (-90...+90°) in Abschnitten von **10°**,

ergibt 18 Abschnitte, gekennzeichnet mit A...R.

Ergeben die beiden ersten Buchstaben im Locator,

erster Buchstabe = Länge, zweiter Buchstabe = Breite,  
Beispiel Reichstag Berlin "JO".

2. **Großfeld** (engl. Square):

Aktuelles Großfeld **Länge** ( $20^\circ$ ) dividiert durch 10 = **2°**,  
ergibt 10 Abschnitte, gekennzeichnet mit 0...9.

Aktuelles Großfeld **Breite** ( $10^\circ$ ) dividiert durch 10 = **1°**,  
ergibt 10 Abschnitte, gekennzeichnet mit 0...9.

Ergeben das dritte (Länge) und vierte Zeichen (Breite) im Locator,  
Beispiel Reichstag Berlin "JO**62**".

3. **Kleinfeld** (engl. Subsquare):

Aktuelles Großfeld **Länge** ( $2^\circ = 120$  Minuten) dividiert durch 24 = **5'**  
ergibt 24 Abschnitte, gekennzeichnet mit a...x (Buchstaben 1 bis 24).

Aktuelles Großfeld **Breite** ( $1^\circ = 60$  Minuten) dividiert durch 24 = **2,5'**  
ergibt 24 Abschnitte, gekennzeichnet mit a...x.

Ergeben das fünfte (Länge) und sechste Zeichen (Breite) im Locator,  
Beispiel Reichstag Berlin "JO6**2qm**".

Funkamateure begnügen sich zumeist mit dieser räumlichen Auflösung. Wenn schon GPS-Systeme auf wenige Meter genau sind, könnte man auch das Feldgitter weiter verfeinern.

4. **Mikrofeld** (engl. Microsquare):


Aktuelles Kleinfeld **Länge** ( $5' = 300$  Sekunden) dividiert durch 10 = **30"**  
ergibt 10 Abschnitte, gekennzeichnet mit 0...9.

Aktuelles Kleinfeld **Breite** ( $2,5' = 150$  Sekunden) dividiert durch 10 = **15"**  
ergibt 10 Abschnitte, gekennzeichnet mit 0...9.


Ergeben das siebte (Länge) und achte Zeichen (Breite) im Locator,  
Beispiel Reichstag Berlin "JO62qm**54**".

Um die Berechnung des Locators zu vereinfachen, werden Längen- und Breitengrade nicht wie oben von  $-180$  bis  $+180^\circ$  bzw.  $-90$  bis  $+90^\circ$  verwendet, sondern von  $0$  bis  $360^\circ$  bzw. von  $0$  bis  $180^\circ$ . Mit der jeweiligen Kantenlänge der rechteckigen Abschnitte in den Feldern 1...4 ist damit eine direkte Zuordnung der Längen- und Breitengrade zu den Buchstaben- und Ziffernfolgen möglich. Am Beispiel Reichstag Berlin im obigen Bild für (1) Großfeld:

- Länge:  $13,376109^\circ$  E =  $+13,376109^\circ$
- Am Bild oben abzulesen:  $+13,37..^\circ$  fällt in das Längengradraster (weiß) von  $0$  bis  $+20^\circ$ , für das der Buchstabe **J** (gelb) steht.
- Berechnung:  $13,376109^\circ$  E  $\Rightarrow 13,376109 + 180 = 193,376109^\circ$  (grünes Feld).  
 Großfeldunterteilung  $20^\circ$   $\Rightarrow 193,376109 / 20 = 9,668805$   
 Ganzzahl aus  $9,668805$   $\Rightarrow 9 \Rightarrow 9 + 1 = 10 \Rightarrow$  zehnter Buchstabe = **J**



- Nachkommazahl von  $9,668805$   $\Rightarrow 0,668805$   
 Großfeldteilung 10  $\Rightarrow 0,668805 * 10 = 6,68805$   
 Ganzzahl aus  $6,68805$   $\Rightarrow$  **6**



- Und so weiter für Länge Kleinfeld (Buchstabe 5, Teilung 24) und Mikrofeld (Ziffer 7, Teilung 10).
- Das Gleiche für die Breite, Buchstaben 2, 6 und Ziffern 4, 8, Großfeldunterteilung 10, weitere Teilungen 10, 24, 10 wie oben.
- Ergebnis



**Das hierzu passende BASCOM-Programm:**

```

'Variables Maidenhead locator -----
Dim sngLng As Single           'Longitude decimal
Dim sngLat As Single          'Latitude decimal
Dim sngGridLen As Single      'WGS84 grid length
Dim bytLocIdx As Byte         'Locator character index
Dim strLocator As String * 8  'Locator (8 characters)
Dim bytLocator(8) As Byte At strLocator Overlay

Sub Maidenhead()
'Calculate Maidenhead locator
'Input  sngLng      Longitude decimal (West negative, East positive)
'       sngLat      Latitude  decimal (South negative, North positive)
'Output strLocator  Locator overlaid with byte array bytLocator()

'Longitude -----
'Step width Field:      20° width 180° West (-180°) to 180° East, 18 segments
'       Square:         2°   (20° / 10)      degrees
'       Subsquare:     5'   (2° = 120' / 24)  arc minutes
'       Microsquare 30'' (5' = 300'' / 10) arc seconds
sngGridLen = sngLng + 180      'Positive longitude 0...360°
Call CalcGrid(0.05)           'Field 0...360° / 20 = A...R
bytLocator(1) = 65 + bytLocIdx '65 = ASCII "A"
Call CalcGrid(10)             'Square 0...20° / 10 = 0...9
bytLocator(3) = 48 + bytLocIdx '48 = ASCII "0"
Call CalcGrid(24)             'Subsquare 0...2° / 24 = A...X
bytLocator(5) = 65 + bytLocIdx
Call CalcGrid(10)             'Microsquare 0...5' / 10 = 0...9
bytLocator(7) = 48 + bytLocIdx

'Latitude -----
'Step width Field:      10° width 90° South (-90°) to 90° North, 18 segments
'       Square:         1°   (10° / 10)      degrees
'       Subsquare:     2.5' (1° = 60' / 24)  arc minutes
'       Microsquare 15'' (2.5' = 150'' / 10) arc seconds
sngGridLen = sngLat + 90      'Positive latitude 0...180°
Call CalcGrid(0.1)           'Field 0...180° / 10 = A...R
bytLocator(2) = 65 + bytLocIdx
Call CalcGrid(10)
bytLocator(4) = 48 + bytLocIdx
Call CalcGrid(24)
bytLocator(6) = 65 + bytLocIdx
Call CalcGrid(10)
bytLocator(8) = 48 + bytLocIdx

End Sub

'*****
Sub CalcGrid(ByVal sngFac As single)
'Calculate new WGS84 grid
'Input  sngFac      Grid factor
'       sngGridLen  Old grid angle segment
'Output bytLocIdx  0=A/0, 1=B/1, 2=C/2 ...Locator Letter/number
'       sngGridLen  Fraction (decimal places) of old angle segment

sngGridLen = sngGridLen * sngFac
bytLocIdx = Int(sngGridLen)
sngGridLen = Frac(sngGridLen)

End Sub

```